include: 'snakepit/pangenie_panel.smk'
include: 'snakepit/pangenie_genotyping.smk'
include: 'snakepit/variant_calling.smk'
include: 'snakepit/variant_accuracy.smk'
include: 'snakepit/association_mapping.smk'
include: 'snakepit/LD.smk'

rule all:
    input:
        ## Pangenome panel creation and genotyping
        'pangenie_panel/multisample-vcfs/graph-filtered.vcf',
        'pangenie_panel/samples.all.pangenie_genotyping_DV.vcf.gz',
        ## Variant comparison between pangenome panel, short reads, and HiFi reads
        'jasmine.vcf',
        expand('SVs/{metric}.gz',metric=('sizes','support','GTs')),
        expand('SNPs/{metric}.csv',metric=('F1','isec')),
        ## Linkage disequilibrium analysis of SVs
        expand('LD/samples.SV.{r2}.{window}.tags.list',r2=list(range(2,10))+[99],window=(10,100,1000)),
        ## Molecular QTL mapping
        expand('QTL/{QTL}/Testis_{variants}/conditionals.01.txt.gz',QTL=('eQTL','sQTL'),variants=('PanGenie','SR'))
from pathlib import PurePath

rule bcftools_annotate:
    input:
        'QTL/PanGenie.vcf.gz'
    output:
        annotation = multiext('LD/annotation.bed.gz','','.tbi'),
        annotated = multiext('LD/variants.vcf.gz','','.tbi'),
        nice = 'LD/nicer_variants.txt'
    threads: 2
    shell:
        '''
        paste <(bcftools query -i 'MAF>0.01' -f '%CHROM\t%POS\t%REF\t%ALT\n' {input}) <(seq 1 $(bcftools index -n {input})) | bgzip -c > {output.annotation[0]}

        tabix -p vcf -b 2 -e 2 {output.annotation[0]}

        zcat {output.annotation[0]} | mawk ' {{ A[$1"_"$2]=$0 }} END {{ for (key in A) {{ print A[key] }} }} ' | cut -f 5 | awk '$1' | sort -n > {output.nice}

        bcftools annotate --threads {threads} -a {output.annotation[0]} -o {output.annotated[0]} -c CHROM,POS,REF,ALT,ID {input}
        tabix -p vcf {output.annotated[0]}
        '''

rule bcftools_query:
    input:
        rules.bcftools_annotate.output['annotated']
    output:
        'LD/SV_IDs.txt'
    shell:
        '''
        bcftools query -i 'abs(ILEN)>=50' -f '%ID\n' {input[0]} > {output}
        '''

rule plink2_LD:
    input:
        vcf = rules.bcftools_annotate.output['annotated'][0],
        nicer_variants = rules.bcftools_annotate.output['nice'],
        tags = rules.bcftools_query.output
    output:
        'LD/samples.SV.{r2}.{window}.tags.list'
    threads: 4
    resources:
        mem_mb = 5000,
        walltime = '60'
    params:
        mem = lambda wildcards, threads, resources: threads*resources.mem_mb,
        out = lambda wildcards, output: PurePath(output[0]).with_suffix('').with_suffix('')
    shell:
        '''
        plink --vcf {input.vcf} --show-tags {input.tags} --tag-r2 0.{wildcards.r2} --tag-kb {wildcards.window} --threads {threads} --memory {params.mem} --chr-set 30 --vcf-half-call h --list-all --out {params.out}
        '''


RepeatMasker --lib ../../../REF_DATA/Repeat_libraries/BosTau9_repeat_library.fasta insertion_sequences.sQTL.fa
 cat insertion_sequences.sQTL.fa.out | grep -v "Simple_repeat" | grep -v "Low_complexity" | /cluster/work/pausch/alex/software/RepeatMasker/util/buildSummary.pl - > insertion_sequences.sQTL.fa.tbl
from pathlib import PurePath

wildcard_constraints:
    _pass = r'permutations|conditionals|nominals',
    chunk = r'\d+',
    chrom = r'\d+',
    MAF = r'\d+',
    vcf = r'(eQTL|gwas)/\S+'

rule normalise_vcf:
    input:
        lambda wildcards: expand(rules.merge_with_population_SR.output,pangenie_mode='genotyping',allow_missing=True) if wildcards.variants == 'PanGenie' else config['small_variants']
    output:
        'QTL/{variants}.vcf.gz'
    threads: 4
    resources:
        mem_mb = 2500,
        disk_scratch = 50
    shell:
        '''
        bcftools norm --threads {threads} -f {config[reference]} -m -any {input} -Ou | \
        bcftools sort -T $TMPDIR -Ou - | \
        bcftools annotate --threads {threads} --set-id '%CHROM\_%POS\_%TYPE\_%REF\_%ALT' -o {output} -
        tabix -fp vcf {output}
        '''

rule exclude_MAF:
    input:
        rules.normalise_vcf.output
    output:
        'QTL/{variants}.exclude_sites.{MAF}.txt'
    shell:
        '''
        bcftools view --threads 2 -Q 0.{wildcards.MAF}:minor -Ou {input} |\
        bcftools query -f '%ID\n'  - > {output}
        '''

def get_pass(_pass,input):
    if _pass == 'permutations':
        return f'--permute {config["permutations"]}'
    elif _pass == 'conditionals':
        return f'--mapping {input.mapping}'
    elif _pass == 'nominals':
        return f'--nominal {config.get("nominal",0.05)}'

rule qtltools_parallel:
    input:
        vcf = rules.normalise_vcf.output,
        exclude = rules.exclude_MAF.output,
        bed = lambda wildcards: config['mol_QTLs'][wildcards.QTL][wildcards.tissue],
        cov = lambda wildcards: config['covariates'][wildcards.QTL][wildcards.tissue],
        mapping = lambda wildcards: 'QTL/{QTL}/{tissue}_{variants}/permutations_all.{MAF}.thresholds.txt' if wildcards._pass == 'conditionals' else []
    output:
        merged = temp('QTL/{QTL}/{tissue}_{variants}/{_pass}.{chunk}.{MAF}.txt.gz')
    params:
        _pass = lambda wildcards,input: get_pass(wildcards._pass,input),
        grp = lambda wildcards: '--grp-best' if wildcards.QTL == 'sQTL' else ''
    threads: 1
    resources:
        mem_mb = 2500,
    shell:
        '''
        QTLtools cis --vcf {input.vcf} --bed {input.bed} --cov {input.cov} --std-err {params._pass} {params.grp} --window {config[window]} --normal --chunk {wildcards.chunk} {config[chunks]} --silent --log /dev/stderr --out /dev/stdout | pigz -p 2 -c > {output}
        '''

rule qtltools_gather:
    input:
        expand(rules.qtltools_parallel.output,chunk=range(0,config['chunks']+1),allow_missing=True)
    output:
        'QTL/{QTL}/{tissue}_{variants}/{_pass}.{MAF}.txt.gz'
    params:
        sort_key = lambda wildcards: '-k9,9n -k10,10n' if wildcards.QTL == 'eQTL' else '-k11,11n -k12,12n'
    localrule: True
    shell:
        '''
        LC_ALL=C; pigz -p 2 -dc {input} | sort --parallel=2 {params.sort_key} | pigz -p 2 -c > {output}
        '''

rule qtltools_FDR:
    input:
        expand(rules.qtltools_gather.output,_pass='permutations',allow_missing=True)
    output:
        'QTL/{QTL}/{tissue}_{variants}/permutations_all.{MAF}.thresholds.txt'
    params:
        out = lambda wildcards, output: PurePath(output[0]).with_suffix('').with_suffix('')
    envmodules:
        'gcc/8.2.0',
        'r/4.2.2'
    shell:
        '''
        Rscript qtltools_runFDR_cis.R {input} 0.05 {params.out}
        '''

rule LD:
    input:
        'QTL/{variants}.vcf.gz'
    output:
        'QTL/{variants}.stats'
    resources:
        mem_mb = 5000
    shell:
        '''
        bcftools annotate -x INFO/AF -o $TMPDIR/sample.vcf.gz {input}
        tabix -p vcf $TMPDIR/sample.vcf.gz
        bcftools view -i 'abs(ILEN)>=50' $TMPDIR/sample.vcf.gz |\
        bcftools stats > {output}
        #awk '/number of records:/ {{ print $6 }} >> {output}
        bcftools +prune -m 0.6 -w 1Mb $TMPDIR/sample.vcf.gz |\
        bcftools stats >> {output}
        #awk '/number of records:/ {{ print $6 }} >> {output}
        '''
from pathlib import PurePath
import subprocess

wildcard_constraints:
    chunk = r'\d*'

chromosomes = list(map(str,range(1,30)))

rule all:				      
    input:
        'pangenie_DV_imputed/samples.all.pangenie_genotyping_DV.imputed.phased.vcf.gz'

rule determine_scatter:
    output:
        temp('scatter.txt')
    params:
        window = config['window'],
        overlap = config['overlap']
    run:
        with open(f'{config["reference"]}.fai') as fai, open(output[0],'w') as fout:
            for line in fai:
                chrom, size = line.split()[:2]
                if chrom not in chromosomes:
                    continue
                for coord in range(1,int(size),params.window):
                    fout.write(f'{chrom}:{max(1,coord-params.overlap)}-{coord+params.window+params.overlap}\n')

checkpoint bcftools_scatter:
    input:
        vcf = 'pangenie/samples.all.pangenie_genotyping_DV.vcf.gz',
        #scatter = 'scatter.txt'
    output:
        temp(directory('pangenie_DV_scatter'))
    threads: 4
    resources:
        mem_mb = 2500
    params:
        scatter = ','.join(chromosomes),
        prefix = 'raw.chr'
    shell:
        '''
        bcftools +scatter {input.vcf} --threads {threads} -s {params.scatter} -Oz -p {params.prefix} -o {output}
        '''

rule beagle4_impute:
    input:
        lambda wildcards: f'pangenie_DV_scatter/{"raw" if wildcards.out == "phased" else "raw"}.chr{{chunk}}.vcf.gz'
    output:
        temp('pangenie_DV_scatter/{out,imputed|phased}.chr{chunk}.vcf.gz')
    threads: 8
    resources:
        mem_mb = 4000,
        walltime = '4:00'
    params:
        prefix = lambda wildcards, output: PurePath(output[0]).with_suffix('').with_suffix(''),
	    ne = 200,
        mem = lambda wildcards, threads, resources: int(resources.mem_mb*threads/1000),
        window = config['window'],
        beagle = config['beagle5']
    shell:
        '''
        java -Xmx{params.mem}g -Xss50m -jar {params.beagle} \
        gt={input} \
        nthreads={threads} \
        out={params.prefix}
        tabix -fp vcf {output}
        '''

def empty_vcf(vcf):
    #hardcode removing the .imputed to get the original vcf filename
    return subprocess.getoutput(f"zgrep -cm1 -v '#' {vcf.replace('.imputed','')}").strip() != "0"

def aggregate_scatter(wildcards):
    checkpoint_dir = checkpoints.bcftools_scatter.get(**wildcards).output[0]
    return sorted([f'pangenie_DV_scatter/phased.chr{chunk}.vcf.gz' for chunk in glob_wildcards(PurePath(checkpoint_dir).joinpath('raw.chr{chunk,\d*}.vcf.gz')).chunk])

rule merge_vcfs:
    input:
        aggregate_scatter
    output:
        'pangenie_DV_imputed/samples.all.pangenie_genotyping_DV.imputed.phased.vcf.gz'
    threads: 2
    resources:
        mem_mb = 2500
    shell:
        '''
        bcftools concat --threads {threads} -a -d exact -o {output} {input}
        tabix -p vcf {output}
        '''

rule beagle_phase_vcf:
    input:
        'pangenie_DV_imputed/samples.all.pangenie_genotyping_DV.imputed.vcf.gz'
    output:
        'pangenie_DV_imputed/samples.all.pangenie_genotyping_DV.imputed.phased.X.vcf.gz'
    params:
        out = lambda wildcards, output: PurePath(output[0]).with_suffix('').with_suffix(''),
        mem = lambda wildcards, threads, resources: int(threads*resources.mem_mb/1024),
        ne = 200
    threads: 12
    resources:
        mem_mb = 10000,
        walltime = '24:00'
    shell:
        '''
        java -Xmx{params.mem}g -jar {config[beagle5]} \
        gt={input} \
        out={params.out} \
        nthreads={threads}
        '''

import os
from pathlib import PurePath

rule all:
    input:
        'GRM/main.hsq'

rule partition_IDs:
    input:
        lambda wildcards: config['vcfs'][wildcards.vcf]
    output:
        SV = 'GRM/{vcf}.SV.ids',
        small = 'GRM/{vcf}.small.ids',
        _all = 'GRM/{vcf}.all.ids'
    shell:
        '''
        bcftools query -i 'abs(ILEN)>=50' -f '%ID\n' {input} > {output.SV}
        bcftools query -e 'abs(ILEN)>=50' -f '%ID\n' {input} > {output.small}
        cat {output.SV} {output.small} > {output._all}
        '''

rule plink_make_bed:
    input:
        vcf = lambda wildcards: config['vcfs'][wildcards.vcf]
    output:
        'bfiles/{vcf}.{chromosome}.bim'
    params:
        bfile = lambda wildcards, output: PurePath(output[0]).with_suffix(''),
        memory = lambda wildcards, threads, resources: int(threads*resources.mem_mb)
    threads: 1
    resources:
        mem_mb = 20000,
        walltime = '15'
    shell:
        '''
        plink2 --native --vcf {input.vcf} --threads {threads} --memory {params.memory} --maf 0.01:minor --cow --make-bed --chr {wildcards.chromosome} --out {params.bfile}
        '''

rule gcta_grm:
    input:
        bfiles = expand('bfiles/{{vcf}}.{chromosome}.bim',chromosome=range(1,30)),
        ids = 'GRM/{vcf}.{variants}.ids' 
    output:
        'GRM/{vcf}.{variants}.grm.bin'
    params:
        _input = lambda wildcards, input: '\\n'.join([str(PurePath(I).with_suffix('')) for I in input.bfiles]),
        _output = lambda wildcards, output: PurePath(output[0]).with_suffix('').with_suffix('')
    threads: 8
    resources:
        mem_mb = 6000
    shell:
        '''
        echo -e "{params._input}" > $TMPDIR/mbfile
        gcta --thread-num {threads} --autosome-num 30 --extract {input.ids} --make-grm-bin --mbfile $TMPDIR/mbfile --out {params._output}
        '''

localrules: prep_covars
rule prep_covars:
    input:
        config['covariates']['eQTL']
    output:
        'GRM/covar.txt'
    shell:
        '''
        awk '{{for(i=1;i<=NF;i++)a[i][NR]=$i}} END {{for(i in a)for(j in a[i])printf"%s"(j==NR?RS:FS),a[i][j]}}' {input} | awk 'NR>1{{print "0",$0}}' > {output}
        '''

localrules: prep_phenotypes
checkpoint prep_phenotypes:
    input:
        config['genes']['eQTL']
    output:
        directory('GRM/molecular_phenotypes')
    shell:
        '''
        mkdir -p {output}
        zcat {input} | awk -v P={output} '{{if (NR==1){{for(i=7;i<=NF;i++)a[i]=$i}} else {{GENE=$4;for(i=7;i<=NF;i++)b[i]=$i;for(i in a)print 0,a[i],b[i] >> P"/"GENE".phen"}} }}'
        '''

def aggregate_phenotypes(wildcards):
      checkpoint_output = checkpoints.prep_phenotypes.get(**wildcards).output[0]
      return expand("GRM/{{vcf}}.{phenotype}.hsq", phenotype=glob_wildcards(os.path.join(checkpoint_output, "{phenotype}.phen")).phenotype)

localrules: gcta_reml
rule gcta_reml:
    input:
        grm = expand('GRM/{{vcf}}.{variants}.grm.bin',variants=('SV','small')),
        covar = rules.prep_covars.output,
        phenotype = 'GRM/molecular_phenotypes/{phenotype}.phen'
    output:
        'GRM/{vcf}.{phenotype}.hsq'
    params:
        grm = lambda wildcards, input: '\\n'.join([PurePath(I).with_suffix('').with_suffix('') for I in input.grm]),
        out = lambda wildcards, output: PurePath(output[0]).with_suffix('')
    shell:
        '''
        gcta --reml --mgrm-bin {params.grm} --pheno {input.phenotype} --qcovar {input.covar} --reml-no-lrt --reml-alg 2 --reml-maxit 10000 --out {params.out}
        '''
#find GRM/molecular_phenotypes/ -type f -name '*.phen' -exec parallel -I@@ -j 6 gcta --reml --grm-bin GRM/main.small --pheno @@ --qcovar GRM/covar.txt --reml-no-lrt --reml-alg 2 --reml-maxit 10000 --out {.}.small ::: {} \+ > /dev/null

localrules: gather_hsq
rule gather_hsq:
    input:
        aggregate_phenotypes
    output:
        'GRM/{vcf,main}.hsq'
    shell:
        '''
        awk '$1=="V(G)/Vp" {{print FILENAME,$2,$3}}' {input} | sed 's/\.hsq//g' > {output}
        '''
#awk '{if($1=="V(G1)/Vp"){a[FILENAME][1]=$2;a[FILENAME][2]=$3}else{if($1=="V(G2)/Vp"){a[FILENAME][3]=$2;a[FILENAME][4]=$3}else{if($3=="V(G)/Vp"){a[FILENAME][5]=$4;a[FILENAME][6]=$5}}}} END {for(F in a){print F,a[F][1],a[F][2],a[F][3],a[F][4],a[F][5],a[F][6]}}' GRM/molecular_phenotypes/*hsq | sed 's/GRM\/molecular_phenotypes\///g;s/\.hsq//g' > GRM/main.hsq


rule gcta_score:
    input:
        expand('bfiles/{{vcf}}.{chromosome}.bim',chromosome=range(1,30))
    output:
        'GRM/{vcf}.score.ld'
    params:
        _input = lambda wildcards, input: '\\n'.join([str(PurePath(I).with_suffix('')) for I in input]),
        _output = lambda wildcards, output: PurePath(output[0]).with_suffix('').with_suffix('')
    threads: 24
    resources:
        mem_mb = 3000,
        walltime = '24:00'
    output:
        ''
    shell:
        '''
        echo -e "{params._input}" > $TMPDIR/mbfile
        gcta --mbfile $TMPDIR/mbfile --ld-score-region 200 --ld-wind 1000 --out {params._output} --thread-num {threads} --autosome-num 30
        '''

rule gcta_stratify:
    input:
        'GRM/{vcf}.score.ld'
    output:
        expand('GRM/{{vcf}}.LD_group{N}.ids',N=range(1,5))
    params:
        _output = lambda wildcards, output: str(PurePath(output[0]).parent)
    shell:
        '''
        Rscript --vanilla {workflow.basedir}/LD_stratify.R {input} {params._output}
        '''
wildcard_constraints:
    pangenie_mode = r'genotyping|phasing'

def get_sample_location(sample):
    fastqs = []
    for R in (1,2):
        fastqs.append(str(Path(f'{config["fastq"]}{sample}_R{R}.fastq.gz').resolve()))
    return fastqs

rule vcfwave:
    input:
        vcf = rules.normalize_vcf.output
    output:
        'pangenie_panel.vcfwave.vcf'
    threads: 8
    resources:
        mem_mb = 1500,
        walltime = '120h'
    shell:
        '''
        vcfwave -t {threads} --quiet {input} > {output}
        '''

rule pangenie_index:
    input:
        reference = config['reference'],
        vcf = rules.vcfwave.output
    output:
        multiext('pangenie_panel','.cereal','.path_segments.fasta')
    params:
        prefix = lambda wildcards, output: PurePath(output[0]).with_suffix('')
    threads: 1
    resources:
        mem_mb = 15000
    shell:
        '''
        pangenie -B {params.prefix} -i /dev/null -r {input.reference} -v {input.vcf} -t {threads}
        '''

rule pangenie_genotype:
    input:
        reference = config['reference'],
        fastq = lambda wildcards: get_sample_location(wildcards.sample),
        pangenie_index = rules.pangenie_index.output
    output:
       'pangenie_panel/{sample}.all.pangenie_genotyping.vcf'
    params:
        prefix = lambda wildcards, output: str(PurePath(output[0]).with_suffix('')).replace(f'_genotyping',''),
        index = lambda wildcards, input: PurePath(input.pangenie_index[0]).with_suffix('')
    threads: 12
    resources:
        mem_mb = 8000,
        scratch = '75G',
        walltime = '4h'
    shell:
        '''
        fastp -w {threads} -i {input.fastq[0]} -I {input.fastq[1]} --stdout -g --thread {threads} --html /dev/null --json /dev/null --dont_eval_duplication | seqtk seq -A > $TMPDIR/reads.fa
        pangenie -i $TMPDIR/reads.fa -r {input.reference} -v {params.index} -t {threads} -j {threads} -s {wildcards.sample} -g -o {params.prefix}
        '''

rule merge_pangenie:
    input:
       vcf = expand('pangenie_panel/{sample}.all.pangenie_{pangenie_mode}.vcf.gz',sample=config['samples'],allow_missing=True),
       tbi = expand('pangenie_panel/{sample}.all.pangenie_{pangenie_mode}.vcf.gz.tbi',sample=config['samples'],allow_missing=True)
    output:
        'pangenie_panel/samples.all.pangenie_{pangenie_mode}.vcf.gz'
    threads: 4
    resources:
        mem_mb = 10000,
        walltime = '24h'
    shell:
        '''
        bcftools merge --threads {threads} -o {output} {input.vcf}
        tabix -p vcf {output}
        '''

rule extract_SVs:
    input:
        rules.merge_pangenie.output
    output:
        'pangenie_panel/samples.all.pangenie_{pangenie_mode}.SVs.vcf.gz'
    threads: 2
    resources:
        mem_mb = 2500
    shell:
        '''
        bcftools view --threads {threads} -i 'abs(ILEN)>50' -o {output} {input}
        '''

rule merge_with_population_SR:
    input:
        DV = config['small_variants'],
        pangenie = rules.merge_pangenie.output
    output:
        'pangenie_panel/samples.all.pangenie_{pangenie_mode}_DV.vcf.gz'
    threads: 8
    resources:
        mem_mb = 2500,
        disk_scratch = 75,
        walltime = '24h'
    shell:
        '''
        bcftools concat -a -D --threads {threads} {input}| \
        grep -v "MONOALLELIC" |\
        bcftools norm --threads {threads} -f {config[reference]} -m -any -Ou - | \
        bcftools norm --threads {threads} -f {config[reference]} -d none -Ou - | \
        bcftools norm --threads {threads} -f {config[reference]} -m +any -Ou - | \
        bcftools sort -T $TMPDIR -Ou - | \
        bcftools annotate --threads {threads} --set-id '%CHROM\_%POS\_%TYPE\_%REF\_%ALT' -o {output} -
        '''

rule compare_pangenie:
    input:
        pangenie = rules.merge_pangenie.output
    output:
        'concordance/{sample}.genotype_concordance_summary_metrics'
    params:
        gc_out = lambda wildcards,output: PurePath(output[0]).parent / f'{wildcards.sample}'
    resources:
        mem_mb = 15000,
        walltime = '60'
    envmodules:
        'gcc/8.2.0',
        'picard/2.25.7'
    shell:
        '''
        picard GenotypeConcordance CALL_VCF={input.pangenie} TRUTH_VCF={config[reference_vcf]} CALL_SAMPLE={wildcards.sample} TRUTH_SAMPLE={wildcards.sample} O={params.gc_out}
        '''
################################ Call variants from haplotype-resolved assemblies ################################
#
# steps:
#
#  1.) align contigs to reference using minimap2
#  2.) determine regions (uniquely) covered contigs haplotypes
#  3.) call variants from assemblies using paftools
#  4.) generate bi-allelic vcf file
#  5.) check mendelian consistency in trios and construct graph (multi-allelic vcf)
#  6.) compute some statistics
#
# output:
#
# variant callset produced from the assemblies (represented as bi-allelic VCF: callset-filtered.vcf)
# pangenome graph produced from the variant calls (represented as multi-allelic VCF: graph-filtered.vcf)
#
##################################################################################################################


configfile: "config.json"
samples = config['assemblies'].keys()
samples_parents = [s for s in samples if not s in config['trios']]
scripts = config['scripts']
outdir = config['outdir']
chromosomes = list(map(str,range(1,30))) #[str(i) for i in range(1,23)] + [config['reference']['prefix'] + 'X', config['reference']['prefix'] + 'Y']
frac_missing = config.get('frac_missing',0.2) # skip positions with more than this fraction of missing alleles

# paftools skips contig-alignments shorter than this threshold
min_alignment_len = 50000

###########################################
#		1) Alignment
###########################################

rule align_assemblies:
    input:
        contigs = lambda wildcards: config['assemblies'][wildcards.sample][int(wildcards.haplotype)],
        reference = config['reference']
    output:
        temp(outdir + "paf/{sample}-hap{haplotype}.sam")
    threads: 8
    resources:
        mem_mb = 8000
    params:
        mm2_opt = config.get('mm2_opt','-x asm20 -m 10000 -z 10000,50 -r 50000 --end-bonus=100 -O 5,56 -E 4,1 -B 5')
    shell:
        '''
        minimap2 -a {params.mm2_opt} --cs -t {threads} {input.reference} {input.contigs} > {output}
        '''

# align assemblies to reference genome
rule align_assemblies_paf:
    input:
        outdir + "paf/{sample}-hap{haplotype}.sam"
    output:
        temp(outdir + "paf/{sample}-hap{haplotype}.paf")
    resources:
        mem_mb = 5000
    shell:
        '''
        paftools.js sam2paf {input} | sort -k6,6 -k8,8n > {output}
        '''


###########################################
#		2) Callable regions
###########################################

# align assemblies to reference and produce BAM output
rule align_assemblies_bam:
    input:
        outdir + "paf/{sample}-hap{haplotype}.sam"
    output:
        outdir + "bam/{sample}-hap{haplotype}.bam"
    threads: 4
    resources:
        mem_mb = 8000,
        disk_scratch = 100
    shell:
        """
        samtools view --threads {threads} -bS {input} | samtools sort --threads {threads} -T $TMPDIR -o {output} --write-index -
        """


# compute regions covered by at least one contig
rule compute_covered_regions:
    input:
        outdir + "bam/{sample}-hap{haplotype}.bam"
    output:
        outdir + "bed/{sample}-hap{haplotype}_covered.bed"
    threads: 1
    resources:
        mem_mb = 5000
    shell:
        "bedtools bamtobed -i {input} | awk '($3-$2) >= {min_alignment_len}' | bedtools merge > {output}"


# compute regions with per-base coverage < 2 
# NOTE: this will NOT remove cases in which there are more than one contig, but all except one contain a deletion)
# 	CCCCCCCCCCCC
# 	C----------C
rule compute_coverage:
    input:
        outdir + "bam/{sample}-hap{haplotype}.bam"
    output:
        outdir + "bed/{sample}-hap{haplotype}_unique.bed"
    threads: 1
    resources:
        mem_mb=5000
    shell:
        "bedtools genomecov -bga -ibam {input} | awk '$4 < 2' | bedtools merge > {output}"


rule intersect_beds:
    input:
        covered= outdir + "bed/{sample}-hap{haplotype}_covered.bed",
        unique= outdir + "bed/{sample}-hap{haplotype}_unique.bed"
    output:
        outdir + "bed/{sample}-hap{haplotype}_callable.bed"
    threads: 1
    resources:
        mem_mb=5000
    shell:
        "bedtools intersect -a {input.covered} -b {input.unique} > {output}"

rule sort_bed:
    input:
        "{filename}.bed"
    output:
        "{filename}-sorted.bed"
    shell:
        "bedtools sort -i {input} > {output}"


rule callable_regions:
    input:
        expand(outdir + "bed/{sample}-hap{haplotype}_callable-sorted.bed", sample=samples, haplotype=[0,1])
    output:
        outdir + "bed/callable-regions.bed"
    params:
        covered = int((1-frac_missing) * len(samples)*2)
    shell:
        "bedtools multiinter -i {input} | awk '$4 > {params.covered}' | bedtools merge > {output}"



##########################################
#		3) Variant Calling 
##########################################

# call variants from alignments
rule paftools:
    input:
        paf= outdir + "paf/{sample}-hap{haplotype}.paf",
        reference = config['reference']
    output:
        temp(outdir + "calls/{sample}-hap{haplotype, [0,1]}.vcf")
    threads: 1
    resources:
        mem_mb=10000
    shell:
        "paftools.js call -L {min_alignment_len} -s {wildcards.sample}_{wildcards.haplotype} -f {input.reference} {input.paf} | sed 's|1/1|1|g' > {output}"


####################################################
#  4) Create bi-allelic VCF and filter it
####################################################


rule compress_vcf:
    input:
        "{filename}.vcf"
    output:
        gz="{filename}.vcf.gz",
        tbi="{filename}.vcf.gz.tbi"
    shell:
        """
        bgzip -c {input} > {output.gz}
        tabix -p vcf {output.gz}
        """

# create a multisample VCF containing all haplotypes
rule collect_all_haplotypes:
    input:
        vcfs=expand("{outdir}calls/{sample}-hap{haplotype}.vcf.gz", outdir=outdir, sample=samples, haplotype=[0,1]),
        tbi=expand("{outdir}calls/{sample}-hap{haplotype}.vcf.gz.tbi", outdir=outdir, sample=samples, haplotype=[0,1])
    output:
        outdir + "calls/all-haplotypes.vcf"
    threads: 2
    resources:
        mem_mb=1000
    shell:
        "bcftools merge --threads {threads} -m none --missing-to-ref {input.vcfs} | python3 {scripts}/assign-variant-ids.py > {output}"


# extract variant ids of callable regions
rule extract_covered_ids:
    input:
        bed = outdir + "bed/{sample}-hap{haplotype}_callable.bed",
        vcf = outdir + "calls/all-haplotypes.vcf"
    output:
        outdir + "bed/{sample}_{haplotype}.txt"
    shell:
        "bedtools intersect -a {input.vcf} -b {input.bed} -wa -f 1.0 | cut -f3 > {output}"


# set alleles outside of callable regions to missing
rule set_to_missing:
    input:
        vcf = outdir + "calls/all-haplotypes.vcf",
        bed = expand("{outdir}bed/{sample}_{haplotype}.txt", outdir=outdir, sample=samples, haplotype=[0,1])
    output:
        outdir + "calls/all-haplotypes-callable.vcf"
    threads: 1
    resources:
        mem_mb=25000,
        walltime = '24:00'
    shell:
        "python3 {scripts}/set-to-missing.py -v {input.vcf} -m {frac_missing} -f {input.bed} > {output}"


# convert haploid VCF into a diploid one by combining haplotypes of each sample
rule write_input:
    output:
        outdir + "samples.txt"
    run:
        with open(output[0], 'w') as txt_output:
            for sample in samples:
                txt_output.write('\t'.join([sample, sample + '_0', sample + '_1']) + '\n')


rule combine_haplotypes:
    input:
        haps = outdir + "calls/all-haplotypes-callable.vcf",
        samples = outdir + "samples.txt"
    output:
        temp(outdir + "multisample-vcfs/callset.vcf")
    threads: 1
    resources:
        mem_mb=10000,
        walltime = '24:00'
    shell:
        'python3 {scripts}/merge_vcfs.py combine_columns -samples {input.samples} -vcf {input.haps} > {output}'


#################################################################
#  5) Check mendelian consistency for trios and construct graph
#################################################################


# generate a file specifying the trio relationships
rule generate_ped_file:
    output:
        "{outdir}trios.ped"
    run:
        with open(output[0], "w") as ped_output:
            for trio in config['trios']:
                father=config['trios'][trio][0]
                mother=config['trios'][trio][1]
                ped_output.write('\t'.join([trio, trio, father, mother]) + '\n')
                
rule generate_samples_file:
    output:
        "{outdir}trio-samples.txt"
    run:
        with open(output[0], "w") as sample_output:
            for trio in config['trios']:
                sample_output.write(trio + '\n')
                for sample in config['trios'][trio]:
                    sample_output.write(sample + '\n')


# remove all variants where there is a mendelian conflict in at least one of the trios
# if no trios are given in config, the vcf does not change.
rule check_mendelian_consistency:
    input:
        vcf="{outdir}multisample-vcfs/callset.vcf",
        ped="{outdir}trios.ped",
        samples="{outdir}trio-samples.txt"
    output:
        vcf="{outdir}multisample-vcfs/callset-filtered.vcf",
        tsv="{outdir}multisample-vcfs/mendelian-consistency.tsv"
    threads: 1
    resources:
        mem_mb=10000,
        walltime = '24:00'
    shell:
        """
        python3 {scripts}/mendelian-consistency.py filter -vcf {input.vcf} -samples {input.samples} -ped {input.ped} -o {output.tsv} > {output.vcf}
        """

rule merge_haplotypes:
    input:
        vcf = outdir + "multisample-vcfs/callset-filtered.vcf",
        reference = config['reference']
    output:
        tmp = temp(outdir + "multisample-vcfs/graph-filtered-tmp-{chr}.vcf")
    #params:
    #    chrom = ','.join([c for c in chromosomes])
    threads: 1
    resources:
        mem_mb=5000,
        walltime = '4:00'
    shell:
        """
        python3 {scripts}/merge_vcfs.py merge -vcf {input.vcf} -r {input.reference} -ploidy 2 -chromosomes {wildcards.chr} -max_edit_distance 0.02 > {output.tmp}
        """

#############################################
#  6) Generate statistics and create plots 
#############################################

rule normalize_vcf:
    input:
        vcfs = expand(outdir + "multisample-vcfs/graph-filtered-tmp-{chr}.vcf.gz",chr=chromosomes),
        reference = config['reference']
    output:
        outdir + "multisample-vcfs/graph-filtered.vcf"
    threads: 2
    resources:
        mem_mb = 5000
    shell:
        "bcftools concat --threads {threads} -Ou {input.vcfs} | bcftools norm --threads {threads} -m +any -d all -f {input.reference} - > {output}"


rule vcfstats_statistics:
    input:
        outdir + "multisample-vcfs/graph-filtered.vcf"
    output:
        txt= outdir + "statistics/vcfstats-stats.txt"
    shell:
        "rtg vcfstats {input} > {output}"


rule untypable_ids:
    input:
        outdir + "multisample-vcfs/callset-filtered.vcf.gz"
    output:
        lists=expand(outdir + "statistics/untypable-ids/{sample}-untypable.tsv", sample=samples_parents),
        summary= outdir + "statistics/untypable-ids.tsv"
    params:
        out= outdir + "statistics/untypable-ids"
    shell:
        "zcat {input} | python3 {scripts}/untypable-ids.py {params.out} > {output.summary}"


rule indel_histogram:
    input:
        outdir + "multisample-vcfs/graph-filtered.vcf"
    output:
        histo= outdir + "statistics/vcftools-stats.indel.hist",
        plot= outdir + "statistics/vcftools-plots/indel-histogram.pdf"
    shell:
        """
        vcftools --vcf {input} --out {outdir}statistics/vcftools-stats --hist-indel-len
        cat {output.histo} | python3 {scripts}/plot-callset-statistics.py length {output.plot} 20000
        """
def read_paths(fname,paths):
    with open(fname,'r') as fin:
        for line in fin:
            parts = line.split()
            chrom = parts[0].split('_')[0]
            if parts[5][0] == '.':
                continue
            
            paths[f'{chrom}_{parts[3][2:]}->{chrom}_{parts[4][2:]}'].add(parts[5].split(':')[0].replace('s',f'{chrom}_'))

from collections import defaultdict
def extract_haplotype_walks(haplotypes):
    paths = defaultdict(set)

    for i in range(1,30):
        for haplotype in haplotypes:
            read_paths(f'ARS_run_5/{haplotype}.path.{i}.L50.bed',paths)

    with open('test.paths','w') as fout:
        for key, v in paths.items():
            fout.write(key +',' +','.join(v) + '\n')


extract_haplotype_walks(['Os1_hifiasm','Os2_hifiasm','Od1_hifiasm','Od2_hifiasm','B31_hifiasm','B32_hifiasm','B41_hifiasm','B42_hifiasm','O_hifiasm','H_hifiasm','P_hifiasm','H_clr','A_clr','S_ont'])
rule pbmm2_align:
    input:
        reads = lambda wildcards: config['hifi_samples'][wildcards.sample]
    output:
        temp(get_dir('pbsv','{sample}.ARS.pbmm2.bam'))
    threads: 24
    resources:
        mem_mb = 3000,
        walltime = '4:00'
    shell:
        'pbmm2 align {config[reference]} {input.reads} {output} --sort --preset CCS -j {threads} --sample {wildcards.sample}'

rule pbsv_discover:
    input:
        get_dir('pbsv','{sample}.ARS.pbmm2.bam')
    output:
        get_dir('pbsv','{sample}.ARS.svsig.gz')
    resources:
        mem_mb = 10000
    shell:
        'pbsv discover {input} {output}'

rule pbsv_call:
    input:
        (get_dir('pbsv','{sample}.ARS.svsig.gz',sample=S) for S in config['hifi_samples'])
    output:
        get_dir('pbsv','samples.pbsv.vcf')
    threads: 8
    resources:
        mem_mb = 3000,
        walltime = "24:00"
    shell:
        'pbsv call --ccs -j {threads} {config[reference]} {input} {output}'

rule pbsv_call_sample:
    input:
        get_dir('pbsv','{sample}.ARS.svsig.gz')
    output:
        get_dir('pbsv','{sample}.pbsv.vcf')
    threads: 8
    resources:
        mem_mb = 5000,
        walltime = "4:00"
    shell:
        'pbsv call --ccs -j {threads} --max-ins-length 100000 {config[reference]} {input} {output}'
 
from pathlib import PurePath

rule all:
    input:
        'SR_SV/SVs.vcf.gz',
        expand('SR_SV/{sample}_{caller}',caller=('insurveyor',),sample=config['HiFi'])

rule picard_add_MQ:
    input:
        '/cluster/work/pausch/inputs/bam/BTA_eQTL/{sample}.bam'
    output:
        multiext('SR_SV/{sample}.bam','','.bai')
    envmodules:
        'gcc/11.4.0',
        'picard/3.1.1'
    threads: 1
    resources:
        mem_mb = 15000,
        walltime = '24h'
    shell:
        '''
        picard FixMateInformation -I {input} -O {output[0]}
        samtools index {output[0]}
        '''

rule insurveyor:
    input:
        rules.picard_add_MQ.output
    output:
        _dir = directory('SR_SV/{sample}_insurveyor'),
        vcf = 'SR_SV/{sample}_insurveyor/out.pass.vcf.gz'
    threads: 4
    resources:
        mem_mb = 7000
    shell:
        '''
        mkdir -p {output._dir}
        python /cluster/work/pausch/alex/software/INSurVeyor/insurveyor.py --threads {threads} --samplename {wildcards.sample} {input[0]} {output._dir} {config[reference]}
        '''

rule SurVClusterer:
    input:
        expand(rules.insurveyor.output['vcf'],sample=config['HiFi'])
    output:
        multiext('SR_SV/cohort.insurveyor','.sv','.vcf.gz','.vcf.gz.tbi')
    params:
        prefix = lambda wildcards, output: PurePath(output[0]).with_suffix('')
    threads: 6
    resources:
        mem_mb = 4000
    shell:
        '''
        awk '$1=="-" {{print $2, "SR_SV/"$2"_insurveyor/out.pass.vcf.gz"}}' config/SR_SV.yaml > $TMPDIR/samples.fofn
        /cluster/work/pausch/alex/software/SurVClusterer/clusterer $TMPDIR/samples.fofn {config[reference]} -t {threads} --min-overlap-precise 0.95 --max-dist-precise 25 --overlap-for-ins -o {params.prefix}
        tabix -fp vcf {output[1]}
        '''

rule delly_call_denovo:
    input:
        '/cluster/work/pausch/inputs/bam/BTA_eQTL/{sample}.bam'
    output:
        'SR_SV/{sample}.delly.denovo.bcf'
    envmodules:
        'gcc/9.3.0',
        'boost/1.74.0',
        'gsl/2.6'
    threads: 4
    resources:
        mem_mb = 4000
    shell:
        '''
        /cluster/work/pausch/alex/software/delly/src/delly call -g {config[reference]} -o {output} {input}
        #bcftools index {output}
        '''

rule delly_merge:
    input:
        expand(rules.delly_call_denovo.output,sample=config['HiFi'])
    output:
        'SR_SV/cohort.delly.denovo.bcf'
    envmodules:
        'gcc/9.3.0',
        'boost/1.74.0',
        'gsl/2.6'
    threads: 4
    resources:
        mem_mb = 2500
    shell:
        '''
        /cluster/work/pausch/alex/software/delly/src/delly merge --precise --pass --minsize 50 --vaf 0.01 -o {output} {input}
        '''

rule delly_call_forced:
    input:
        bam = '/cluster/work/pausch/inputs/bam/BTA_eQTL/{sample}.bam',
        panel = rules.delly_merge.output
    output:
        'SR_SV/{sample}.delly.forced.vcf.gz'
    envmodules:
        'gcc/9.3.0',
        'boost/1.74.0',
        'gsl/2.6'
    threads: 4
    resources:
        mem_mb = 4000
    shell:
        '''
        /cluster/work/pausch/alex/software/delly/src/delly call -g {config[reference]} -v {input.panel} -o $TMPDIR/sample.bcf {input.bam}
        bcftools view -o {output} $TMPDIR/sample.bcf
        tabix -fp vcf {output}
        '''

rule bcftools_merge:
    input:
        lambda wildcards: expand(rules.delly_call_forced.output,sample=config['HiFi']) if wildcards.caller == 'delly.forced' else (expand(rules.survtyper.output['vcf'],sample=config['HiFi']) if wildcards.caller == 'insurveyor.forced' else expand(rules.insurveyor.output['vcf'],sample=config['HiFi']))
    output:
        'SR_SV/cohort.{caller}.vcf.gz'
    resources:
        mem_mb = 10000
    shell:
        '''
        bcftools merge -m id -o {output} {input}
        tabix -fp vcf {output}
        '''

rule delly_filter:
    input:
        expand(rules.bcftools_merge.output,caller='delly.forced')
    output:
        'SR_SV/cohort.delly.filtered.vcf.gz'
    envmodules:
        'gcc/9.3.0',
        'boost/1.74.0',
        'gsl/2.6'
    threads: 1
    resources:
        mem_mb = 2500
    shell:
        '''
        /cluster/work/pausch/alex/software/delly/src/delly filter -f germline --altaf 0.01 --minsize 50 --pass -o $TMPDIR/sample.bcf {input}
        bcftools view -o {output} $TMPDIR/sample.bcf
        tabix -fp vcf {output}
        '''

#awk '$1=="-" {print $2, "SR_SV/"$2"_insurveyor/out.pass.vcf.gz"}' ../config/SR_SV.yaml > ../insurveyor.fofn
rule survtyper:
    input:
        vcf = rules.SurVClusterer.output[1],
        bam = '/cluster/work/pausch/inputs/bam/BTA_eQTL/{sample}.bam'
    output:
        _dir = directory('SR_SV/{sample}_insurveyor_forced'),
        vcf = 'SR_SV/{sample}_insurveyor_forced/genotyped.vcf.gz'
    threads: 6
    resources:
        mem_mb = 8000,
        walltime = '4h'
    shell:
        '''
        python /cluster/work/pausch/alex/software/SurVTyper/survtyper.py --threads {threads} --samplename {wildcards.sample} {input.vcf} {input.bam} {output._dir} {config[reference]}
        '''

rule merge_callers:
    input:
        delly = rules.delly_filter.output,
        insurveyor = expand(rules.bcftools_merge.output,caller='insurveyor.forced')
    output:
        'SR_SV/SVs.vcf.gz'
    threads: 2
    shell:
        '''
        bcftools view {input.delly} -e "SVTYPE=='INS'" -o $TMPDIR/deldel.vcf.gz
        tabix -p vcf $TMPDIR/deldel.vcf.gz
        bcftools concat -a --threads 2 -D -o $TMPDIR/concat.vcf.gz $TMPDIR/deldel.vcf.gz {input.insurveyor}
        tabix -p vcf $TMPDIR/concat.vcf.gz
        bcftools view -e INFO/INCOMPLETE_ASSEMBLY!=0 -o {output} $TMPDIR/concat.vcf.gz 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29
        tabix -p vcf {output}
        '''
from pathlib import PurePath

## Add reference path for happy container
workflow._singularity_args = f'-B $TMPDIR -B {PurePath(config["reference"]).parent}'

def get_variants(variant,caller):
    input_dict = {
              'SNPs_PG':'variant_calling/panel.small.vcf.gz',
              'SNPs_DV':config['small_variants'],
              'SVs_PG': 'variant_calling/panel.SV.vcf',
              'SVs_Sniffles': 'variant_calling/samples.denovo.sniffles.vcf.gz'
              }
    return input_dict[f'{variant}_{caller}']

rule split_vcf:
    input:
        lambda wildcards: get_variants(wildcards.variant,wildcards.caller)#input_dict[f'{wildcards.variant}_{wildcards.caller}']
    output:
        multiext('{variant,SVs|SNPs}/{sample}.{caller,PG|DV|Sniffles}.vcf.gz','','.tbi')
    params:
        regions = ' '.join(map(str,range(1,30))),
        SVs = lambda wildcards: {'SVs_Sniffles':"-i 'F_MISSING<0.2&&abs(ILEN)>=50&&INFO/SVTYPE!=\"BND\"'",'SVs_PG':"-i 'F_MISSING<0.2&&abs(ILEN)>=50'"}.get(f'{wildcards.variant}_{wildcards.caller}',"-i 'F_MISSING<0.2'"),
        sample = lambda wildcards: f'-s {wildcards.sample}' if wildcards.sample != 'all' else ''
    shell:
        '''
        bcftools view -c 1 -a {params.SVs} {params.sample} -o {output[0]} {input} {params.regions}
        tabix -p vcf {output[0]}
        '''

rule bcftools_isec:
    input:
        vcf_truth = expand(rules.split_vcf.output,caller='DV',variant='SNPs',allow_missing=True),
        vcf_query = expand(rules.split_vcf.output,caller='PG',variant='SNPs',allow_missing=True)
    output:
        'SNPs/{sample}.isec'
    shell:
        '''
        bcftools isec -n +1 {input.vcf_truth[0]} {input.vcf_query[0]} | awk ' {{ A[$5]+=1 }} END {{ print "{wildcards.sample}",A["01"],A["10"],A["11"] }}' > {output}
        '''

rule gather_isec:
    input:
        expand(rules.bcftools_isec.output,sample=config['samples'])
    output:
        'SNPs/isec.csv'
    localrule: True
    shell:
        '''
        {{ echo "sample PG DV Mutual" ; cat {input} ; }} > {output} 
        '''

rule happy:
    input:
        vcf_truth = expand(rules.split_vcf.output,caller='DV',variant='SNPs',allow_missing=True),
        vcf_query = expand(rules.split_vcf.output,caller='PG',variant='SNPs',allow_missing=True),
        reference = config['reference']
    output:
        csv = 'SNPs/{sample}.summary.csv',
        others = temp(multiext('SNPs/{sample}','.bcf','.bcf.csi','.extended.csv','.roc.all.csv.gz','.runinfo.json'))
    params:
        _dir = lambda wildcards, output: PurePath(output.csv).with_suffix('').with_suffix('')
    container: 'docker://pkrusche/hap.py'
    threads: 2
    resources:
        mem_mb = 5000,
        scratch = '10G'
    shell:
        '''
        /opt/hap.py/bin/hap.py -r {input.reference} --bcf --usefiltered-truth --no-roc --no-json -L --pass-only --scratch-prefix $TMPDIR -X --threads {threads} -o {params._dir} {input.vcf_truth[0]} {input.vcf_query[0]}
        '''

rule gather_SR_happy:
    input:
        expand(rules.happy.output[0],sample=config['samples'])
    output:
        'SNPs/F1.csv'
    localrule: True
    shell:
        '''
        echo -e "variant truth query recall precision truth_TiTv query_TiTv sample" > {output}
        for i in {input}
        do
          awk -v I=$(basename $i) -F',' '$2=="PASS" {{ split(I,a,"."); print $1,$3,$6,$10,$11,$14,$15,a[1] }}' $i >> {output}
        done
        '''

rule jasmine:
    input:
        vcfs = expand(rules.split_vcf.output,caller=('PG','Sniffles'),variant='SVs',allow_missing=True)
    output:
        'SVs/{sample}.jasmine.txt'
    params:
        _input = lambda wildcards, input: ','.join(input.vcfs).replace('.gz','').replace('SVs','$TMPDIR/SVs')
    conda:
        'jasmine'
    threads: 1
    resources:
        mem_mb= 5000,
        walltime = '60',
        scratch = '5G'
    shell:
        '''
        mkdir -p $TMPDIR/SVs
        pigz -dc {input.vcfs[0]} > $TMPDIR/SVs/{wildcards.sample}.PG.vcf
        pigz -dc {input.vcfs[1]} > $TMPDIR/SVs/{wildcards.sample}.Sniffles.vcf

        jasmine --comma_filelist file_list={params._input} threads={threads} out_file=/dev/stdout out_dir=$TMPDIR \
        genome_file={config[reference]} --pre_normalize --ignore_strand --allow_intrasample --ignore_type \
        max_dist_linear=1 max_dist=1000 > {output}
        #|\
                #grep -hoP "SUPP_VEC=\K\d+" | awk ' {{ A[$1]+=1 }} END {{ print "{wildcards.sample}",A["01"],A["10"],A["11"] }}' > {output}
        '''

rule gather_jasmine:
    input:
        expand(rules.jasmine.output[0],sample=config['HiFi_samples'])
    output:
        'SVs/jasmine.csv'
    localrule: True
    shell:
        '''
        {{ echo "sample Sniffles PG Mutual" ; cat {input} ; }} > {output}
        '''
rule minimap2_align:
    input:
        lambda wildcards: config['HiFi_samples'][wildcards.sample]
    output:
        bam = temp('alignments/{sample}.HiFi.bam'),
        csi = temp('alignments/{sample}.HiFi.bam.csi')
    threads: 12
    resources:
        mem_mb = 6000,
        disk_scratch = 100,
        walltime = '24:00'
    shell:
        '''
        minimap2 -axmap-hifi -t {threads} {config[reference]} {input} |  samtools sort - -m 3000M -@ 4 -T $TMPDIR --write-index -o {output.bam}
        '''

rule sniffles_call:
    input:
        bam = 'alignments/{sample}.HiFi.bam'
    output:
        vcf = temp('variant_calling/{sample}.sniffles.vcf.gz'),
        snf = temp('variant_calling/{sample}.sniffles.snf')
    threads: 4
    resources:
        mem_mb = 2500
    conda:
        'sniffles'
    shell:
        '''
        sniffles --input {input.bam} --reference {config[reference]} --sample-id {wildcards.sample} --threads {threads} --vcf {output.vcf} --snf {output.snf}
        '''

rule sniffles_merge:
    input:
        snfs = expand('variant_calling/{sample}.sniffles.snf',sample=config['HiFi_samples'])
    output:
        vcf = 'variant_calling/samples.sniffles.vcf.gz'
    threads: 2
    resources:
        mem_mb = 2500
    conda:
        'sniffles'
    shell:
        '''
        sniffles --input {input.snfs} --reference {config[reference]} --threads {threads} --vcf {output.vcf}
        '''

rule bcftools_autosomes:
    input:
        'variant_calling/samples.sniffles.vcf.gz'
    output:
        'variant_calling/samples.sniffles.autosomes.vcf'
    threads: 2
    resources:
        mem_mb = 4000
    params:
        regions = ','.join(map(str,range(1,30)))
    shell:
        '''
        bcftools view --threads {threads} -r {params.regions} -o {output} {input}
        '''

rule bcftools_stuff:
    input:
        rules.bcftools_autosomes.output
    output:
        sizes = 'SVs/sizes.gz',
        support = 'SVs/support.gz',
        GTs = 'SVs/GTs.gz'
    localrule: True
    shell:
        '''
        bcftools query -e 'INFO/SVTYPE=="BND"' -f '%INFO/SVLEN\\n' {input} | pigz -p2 -c > {output.sizes}
        bcftools query -e 'INFO/SVTYPE=="BND"' -f '%INFO/SUPP_VEC\\n' {input} | mawk ' {{ print gsub(1,2,$1) }} ' | pigz -p2 -c > {output.support}
        bcftools query -e 'INFO/SVTYPE=="BND"' -f '[%GT ]\\n' {input} | sed 's"|" "g' | sed 's"/" "g' | sed 's/\./nan/g' | pigz -p2 -c > {output.GTs}
        '''

rule bcftools_split_panel:
    input:
        expand(rules.merge_pangenie.output,pangenie_mode='genotyping',allow_missing=True)
    output:
        SV = 'variant_calling/panel.SV.vcf',
        small = 'variant_calling/panel.small.vcf.gz'
    params:
        SV_size = 50,#config['SV_size'],
        bcf = '$TMPDIR/normed.bcf'
    threads: 2
    resources:
        mem_mb = 4000,
        disk_scratch = 10
    shell:
        '''
        bcftools norm --threads {threads} -f {config[reference]} -m -any -Ou {input} > {params.bcf}
        bcftools view -i 'abs(ILEN)>={params.SV_size}' -o {output.SV} {params.bcf}
        bcftools view -e 'abs(ILEN)>={params.SV_size}' -o {output.small} {params.bcf}
        tabix -fp vcf {output.small}
        '''

rule jasmine_intersect:
    input:
        read = 'variant_calling/samples.sniffles.autosomes.vcf',
        asm = 'variant_calling/panel.SV.vcf'
    output:
        'jasmine.vcf',
        'jasmine_overlaps.txt'
    params:
        _input = lambda wildcards, input: ','.join(input)
    conda:
        'jasmine'
    threads: 2
    resources:
        mem_mb = 3000,
        disk_scratch = 5
    shell:
        '''
        jasmine --comma_filelist file_list={params._input} threads={threads} out_file={output[0]} out_dir=$TMPDIR spec_reads=0 genome_file={config[reference]} min_seq_id=.5 --pre_normalize --ignore_strand --allow_intrasample --normalize_type
        grep -vE "SVTYPE=(INV|TRA)" {output[0]} | grep -oP "(SVLEN=-?\d*|SUPP_VEC=\d{{2}})" | sed 's/[A-Z,=,_]*//g'  | paste -s -d' \n' > {output[1]}
        '''
#jasmine --comma_filelist file_list=smoove_SV/All_filter_type.vcf,eQTL_GWAS/variants/variant_calling/panel.SV.vcf threads=1 out_file=SR_LR.vcf out_dir=$TMPDIR spec_reads=0 genome_file=REF_DATA/ARS-UCD1.2_Btau5.0.1Y.fa min_seq_id=0 --pre_normalize --ignore_strand --allow_intrasample max_dist_linear=1 --normalize_type --dup_to_ins
#bcftools query -f '%CHROM\t%POS\t%INFO/END\t%INFO/SVTYPE\t0\t+\t%POS\t%INFO/END\t%INFO/SUPP_VEC\n' SR_LR.vcf | grep -vE "(INV|TRA)"  | sed s'/10$/50,200,50/g' | sed s'/11$/250,100,150/g' | sed s'/01$/50,20,250/g' >> SR_LR.bed 

## Assembly rules:
rule minimap2_align_asm:
    input:
        lambda wildcards: config['asm'][wildcards.sample]
    output:
        'alignments/{sample}.asm.paf'
    threads: 4
    resources:
        mem_mb = 10000
    shell:
        '''
        minimap2 -cx asm5 -t {threads} --cs {config[reference]} {input} > {output}
        '''

rule paftools_call:
    input:
        'alignments/{sample}.asm.paf'
    output:
        'variant_calling/{sample}.asm.paf'
    threads: 1
    resources:
        mem_mb = 2000
    shell:
        '''
        sort -k6,6 -k8,8n {input} | paftools.js call -f {config[reference]} {input} > {output.vcf}
        '''

